第七章 拟合和逼近

六、例

设给定函数

$$f(x) = x - e^{-x}$$

从 $x_1 = 0$ 开始，取步长 $h = 0.1$ 的 20 个数据点，求 5 次最小二乘拟合多项式

$$P_6(x) = a_1 + a_2(x - \overline{x}) + a_3(x - \overline{x})^2 + \cdots + a_6(x - \overline{x})^5$$

其中

$$\overline{x} = \sum_{i=1}^{20} x_i / 20 = 0.95$$

在本问题中，N＝20，M＝6。



最小二乘曲线拟合

一、功能

用最小二乘法求 n 个数据点的拟合多项式。

二、方法说明

设已知 n 个数据点 $(x_i, y_i)(i = 1,2,\cdots,n)$，要求 $(m-1)$ 次最小二乘拟合多项式

$$P_m(x) = a_1 + a_2 x + a_3 x^2 + \cdots + a_m x^{m-1}$$

用幂函数

其中 $m \leqslant n$ 且 $m \leqslant 20$。

设拟合多项式为各正交多项式 $Q_j(x)$ $(j = 1,2,\cdots,m)$ 的线性组合，即

$$P_m(x) = q_1 Q_1(x) + q_2 Q_2(x) + \cdots + q_m Q_m(x)$$

其中 $Q_j(x)$ $(j = 1,2,\cdots,m)$ 由下列递推公式构造：

$$Q_1(x) = 1$$

$$Q_2(x) = (x - \alpha_2)$$

$$Q_{j+1}(x) = (x - \alpha_{j+1})Q_j(x) - \beta_j Q_{j-1}(x),\ j = 2,3,\cdots,m-1$$

若设

$$d_j = \sum_{i=1}^{n} Q_j^2(x_i),\ j = 1,2,\cdots,m$$

则

$$\alpha_{j+1} = \sum_{i=1}^{n} x_i Q_j^2(x_i)/d_j,\ j = 1,2,\cdots,m-1$$

$$\beta_j = d_j/d_{j-1},\qquad j = 2,3,\cdots,m-1$$

可以证明，由上面构造的多项式 $\{Q_j\}(j=1,2,\cdots,m)$ 是互相正交的。根据最小二乘原理，可得

$$q_j = \sum_{i=1}^{n} y_i Q_j(x_i)/d_j,\ j = 1,2,\cdots,m$$

最后可以化成一般的 $m-1$ 次多项式

$$P_m(x) = a_1 + a_2 x + a_3 x^2 + \cdots + a_m x^{m-1}$$

具体计算步骤如下。

(1) $Q_1(x) = 1$，可得

$$b_1 = 1,\ d_1 = n,\ q_1 = \sum_{i=1}^{n} y_i/d_1,\ \alpha = \sum_{i=1}^{n} x_i/d_1$$

$$a_1 = q_1 b_1$$

(2) $Q_2(x) = x - \alpha$，可得

$$t_2 = 1,\ t_1 = -\alpha$$

$$d_2 = \sum_{i=1}^{n} Q_2^2(x_i),\ q_2 = \sum_{i=1}^{n} y_i Q_2(x_i)/d_2$$

$$\alpha = \sum_{i=1}^{n} x_i Q_1^2(x_i)/d_2,\ \beta = d_2/d_1$$

$$a_2 = q_2 t_2,\ q_2 t_1 + a_1 \Rightarrow a_1$$

(3) 对于 $j = 3,4,\cdots,m$，作以下各步：

$$\begin{aligned} Q_j(x) &= (x-\alpha)Q_{j-1}(x) + \beta Q_{j-2}(x) \\ &= (x-\alpha)(t_{j-1}x^{j-2} + \cdots + t_2 x + t_1) \\ &\quad - \beta(b_{j-2}x^{j-3} + \cdots + b_2 x + b_1) \\ &\triangleq s_j x^{j-1} + s_{j-1}x^{j-2} + \cdots + s_2 x + s_1 \end{aligned}$$

其中 s_k 由下列递推公式计算：

$$\begin{cases} s_j = t_{j-1} \\ s_{j-1} = -\alpha\, t_{j-1} + t_{j-2} \\ s_k = -\alpha\, t_k + t_{k-1} - \beta b_k,\ k = j-2,\cdots,2 \\ s_1 = -\alpha\, t_1 - \beta b_1 \end{cases}$$

再计算

$$d_j = \sum_{i=1}^{n} Q_j^2(x_i), \quad q_j = \sum_{i=1}^{n} y_i Q_j(x_i)/d_j$$

$$\alpha = \sum_{i=1}^{n} x_i Q_j^2(x_i)/d_j, \quad \beta = d_j/d_{j-1}$$

由此可以计算相应的 a_k：

$$\begin{cases} a_j = q_j s_j \\ a_k + q_j s_k \Rightarrow a_k, \quad k = j-1, \cdots, 1 \end{cases}$$

且

$$\begin{cases} t_j = s_j \\ b_k = t_k, \quad t_k = s_k, \quad k = j-1, \cdots, 1 \end{cases}$$

在实际计算过程中，为了防止运算溢出，x_i 用

$$x_i' = x_i - \overline{x}, i = 1, 2, \cdots, n$$

代替。其中

$$\overline{x} = \sum_{i=1}^{n} x_i / n$$

此时，拟合多项式的形式为

$$P_m(x) = a_1 + a_2(x - \overline{x}) + a_3(x - \overline{x})^2 + \cdots + a_m(x - \overline{x})^{m-1}$$

实现此功能的子程序如下：

三、子程序语句

```
SUBROUTINE HPIR1(X,Y,A,N,M,DT1,DT2,DT3)
```

四、形参说明

X,Y——均为双精度实型一维数组，长度为N，输入参数。分别存放N个数据点的X坐标与Y坐标。

A——双精度实型一维数组，长度为M，输出参数。存放M−1次拟合多项式的M个系数，即

$$P_M(x) = a_1 + a_2(x-\bar{x}) + a_3(x-\bar{x})^2 + \cdots + a_M(x-\bar{x})^{M-1}$$

N——整型变量，输入参数。数据点数。

M——整型变量，输入参数。拟合多项式的项数，即次数为M−1，要求M≤N且M≤20。

DT1,DT2,DT3——均为双精度实型变量，输出参数。分别为拟合多项式与数据点偏差的平方和、绝对值之和、绝对值的最大值。

```
PROGRAM EXAMPLE08
DIMENSION X(20),Y(20),A(6)
    DOUBLE PRECISION X,Y,A,DT1,DT2,DT3,B
    B=0.0
    DO 10 I=1,20
       X(I)=B+(I-1)*0.1
       Y(I)=X(I)-EXP(-X(I))
10  CONTINUE
    N=20
    M=6
    CALL HPIR1(X,Y,A,N,M,DT1,DT2,DT3)
    WRITE(*,*)
    WRITE(*,20) (I,A(I),I=1,M)
20  FORMAT(1X,'A(',I2,' )=',D15.6)
    WRITE(*,*)
    WRITE(*,30) DT1,DT2,DT3
30  FORMAT(1X,'DT1=',D12.6,5X,'DT2=',D12.6,5X,'DT3=',D12.6)
    END
```

运行结果为

```
A( 1 )=     .563248D+00
A( 2 )=     .138675D+01
A( 3 )=    -.193134D+00
A( 4 )=     .644035D-01
A( 5 )=    -.168412D-01
A( 6 )=     .334429D-02

DT1=  .180174D-08     DT2=  .168505D-03     DT3=  .153940D-04
```

0.563248+1.38675*(x-0.95)-0.193134*(x-0.95)^2+0.0644035*(x-0.95)^3-0.0168412*(x-0.95)^4+0.00334429*(x-0.95)^5



```
SUBROUTINE HPIR1(X,Y,A,N,M,DT1,DT2,DT3)
      DIMENSION X(N),Y(N),A(M),S(20),T(20),B(20)
      DOUBLE PRECISION X,Y,A,S,T,B,DT1,DT2,DT3, &
                           Z,D1,P,C,D2,G,Q,DT
      DO 5 I=1,M
5     A(I)=0.0
      IF (M.GT.N) M=N
      IF (M.GT.20) M=20
      Z=0.0
      DO 10 I=1,N
10    Z=Z+X(I)/N
      B(1)=1.0
      D1=N
      P=0.0
      C=0.0
```

```
      DO 20 I=1,N
         P=P+(X(I)-Z)
         C=C+Y(I)
   20 CONTINUE
      C=C/D1
      P=P/D1
      A(1)=C*B(1)
      IF (M.GT.1) THEN
         T(2)=1.0
         T(1)=-P
         D2=0.0
         C=0.0
         G=0.0
         DO 30 I=1,N
            Q=X(I)-Z-P
```

```
        D2=D2+Q*Q
        C=Y(I)*Q+C
        G=(X(I)-Z)*Q*Q+G
30    CONTINUE
      C=C/D2
      P=G/D2
      Q=D2/D1
      D1=D2
      A(2)=C*T(2)
      A(1)=C*T(1)+A(1)
    END IF
    DO 100 J=3,M
      S(J)=T(J-1)
      S(J-1)=-P*T(J-1)+T(J-2)
      IF (J.GE.4) THEN
```

```
            DO 40 K=J-2,2,-1
40          S(K)=-P*T(K)+T(K-1)-Q*B(K)
        END IF
        S(1)=-P*T(1)-Q*B(1)
        D2=0.0
        C=0.0
        G=0.0
        DO 70 I=1,N
            Q=S(J)
            DO 60 K=J-1,1,-1
60          Q=Q*(X(I)-Z)+S(K)
            D2=D2+Q*Q
            C=Y(I)*Q+C
            G=(X(I)-Z)*Q*Q+G
70      CONTINUE
```

```
      C=C/D2
      P=G/D2
      Q=D2/D1
      D1=D2
      A(J)=C*S(J)
      T(J)=S(J)
      DO 80 K=J-1,1,-1
        A(K)=C*S(K)+A(K)
        B(K)=T(K)
        T(K)=S(K)
80    CONTINUE
100     CONTINUE
    DT1=0.0
    DT2=0.0
    DT3=0.0
```

```
      DO 120 I=1,N
         Q=A(M)
         DO 110 K=M-1,1,-1
110         Q=Q*(X(I)-Z)+A(K)
         DT=Q-Y(I)
         IF (ABS(DT).GT.DT3) DT3=ABS(DT)
         DT1=DT1+DT*DT
         DT2=DT2+ABS(DT)
120   CONTINUE
      RETURN
      END
```

六、例

取函数 $f(x) = \mathrm{arctg}x$ 在区间[-1,1]上的 101 个等距点

$$x_i = -1.0 + (i-1) * 0.02, i = 1,2,\cdots,101$$

上的函数值

$$y_i = f(x_i), i = 1,2,\cdots,101$$

根据此 101 个数据点构造切比雪夫意义下的五次拟合多项式

$$P(x) = a_1 + a_2 x + a_3 x^2 + a_4 x^3 + a_5 x^4 + a_6 x^5$$

在本例中，N=101，M=6，M1=7。



切比雪夫曲线拟合

一、功能

给定 n 个数据点，求切比雪夫(Chebyshev)意义下的最佳拟合多项式。

二、方法说明

设给定 n 个数据点

$$(x_i, y_i), i = 1, 2, \cdots, n$$

其中 $x_1 < x_2 < \cdots < x_{n-1} < x_n$。求 $m-1(m<n)$ 次多项式

$$P(x) = a_1 + a_2 x + \cdots + a_m x^{m-1}$$

使在 n 个给定点上的偏差最大值最小，即

$$\max_{1 \leqslant i \leqslant n} |P(x_i) - y_i| = \min$$

其计算步骤如下。

从给定的 n 个数据点的自变量值 $x_i(i = 1, 2, \cdots, n)$ 中选取 $m+1$ 个不同点 $u_1, u_2, \cdots, u_{m+1}$ 组成初始参考点集。

设在初始点集 $u_1, u_2, \cdots, u_{m+1}$ 上，参考多项式 $\Phi(x)$ 的偏差为 h，即参考多项式 $\Phi(x)$ 在初始点集上的取值为

$$\Phi(u_i) = f(u_i) + (-1)^i h, i = 1, 2, \cdots, m+1$$

且 $\Phi(u_i)$ 的各阶差商是 h 的线性函数。

由于 $\Phi(x)$ 为 $m-1$ 次多项式，其 m 阶差商等于零，由此可以求出 h。再根据 $\Phi(u_i)$ 的各阶差商，由牛顿插值公式可求出 $\Phi(x)$：

$$\Phi(x) = a_1 + a_2 x + a_3 x^2 + \cdots + a_m x^{m-1}$$

令

$$hh=\max_{1\leqslant i\leqslant n}\left|\Phi(x_i)-y_i\right|$$

若 $hh=h$，则 $\Phi(x)$ 即为所求的拟合多项式。

若 $hh>h$，则用达到偏差最大值的点 x_j 代替点集$\{u_j\}(i=1,2,\cdots,m+1)$ 中离 x_j 最近、且具有与$(\Phi(x_j)-y_j)$的符号相同的点，从而构造一个新的参考点集。用这个新的参考点集重复以上过程，直到最大逼近误差等于参考偏差为止。

实现此功能的子程序如下：

三、子程序语句

SUBROUTINE HCHIR(X,Y,N,A,M,M1)

四、形参说明

X,Y——均为双精度实型一维数组，长度为N，输入参数。存放N个数据点的X坐标与Y坐标。

N——整型变量，输入参数。数据点数。

A——双精度实型一维数组，长度为M1=M+1，输出参数。其中A(i) (i=1,2,…,M)存放M−1次拟合多项式

$$P(x) = a_1 + a_2x + a_3x^2 + \cdots + a_mx^{m-1}$$

的系数a_i；A(M+1)存放拟合多项式的偏差。若A(M+1)为负值，则说明在迭代过程中参考偏差绝对值不再增大，其绝对值为当前选择的参考偏差的绝对值。

M——整型变量，输入参数。拟合多项式的项数，即次数为M−1。要求M<N，且M≤19。

M1——整型变量，输入参数。M1=M+1。

```
PROGRAM EXAMPLES081
DIMENSION X(101),Y(101),A(7)
DOUBLE PRECISION X,Y,A
   N=101
   M=6
   M1=7
   DO 10 I=1,N
     X(I)=-1.0+(I-1)*0.02
     Y(I)=ATAN(X(I))
10 CONTINUE
   CALL HCHIR(X,Y,N,A,M,M1)
   WRITE(*,*)
   WRITE(*,20) (I,A(I),I=1,M)
   WRITE(*,*)
   WRITE(*,30) A(M1)
   WRITE(*,*)
20 FORMAT(1X,'A(',I2,' )=',D15.6)
30 FORMAT(1X,'HMAX=',D15.6)
   END
```

运行结果为

A(1)=　−.430478D−11

A(2)=　.995364D+00

A(3)=　.548259D−11

A(4)=　−.288716D+00

A(5)=　−.372085D−10

A(6)=　.793575D−01

HMAX=　.607072D−03



拟合多项式为

0.995364*x-0.288716*x^3+0.079357*x^5

```
SUBROUTINE HCHIR(X,Y,N,A,M,M1)
      DIMENSION X(N),Y(N),A(M1),IX(20),H(20)
      DOUBLE PRECISION X,Y,A,H,HA,HH,Y1,Y2,H1,H2,D,HM
      DO 5 I=1,M1
5     A(I)=0.0
      IF (M.GE.N) M=N-1
      IF (M.GE.20) M=19
      M1=M+1
      HA=0.0
      IX(1)=1
      IX(M1)=N
      L=(N-1)/M
      J=L
      DO 10 I=2,M
         IX(I)=J+1
```

```
      J=J+L
10  CONTINUE
20  HH=1.0
    DO 30 I=1,M1
      A(I)=Y(IX(I))
      H(I)=-HH
      HH=-HH
30  CONTINUE
    DO 50 J=1,M
      II=M1
      Y2=A(II)
      H2=H(II)
      DO 40 I=J,M
        D=X(IX(II))-X(IX(M1-I))
        Y1=A(M-I+J)
```

```
        H1=H(M-I+J)
        A(II)=(Y2-Y1)/D
        H(II)=(H2-H1)/D
        II=M-I+J
        Y2=Y1
        H2=H1
40    CONTINUE
50  CONTINUE
    HH=-A(M1)/H(M1)
    DO 60 I=1,M1
60  A(I)=A(I)+H(I)*HH

    DO 80 J=1,M-1
      II=M-J
      D=X(IX(II))
```

```
      Y2=A(II)
      DO 70 K=M1-J,M
        Y1=A(K)
        A(II)=Y2-D*Y1
        Y2=Y1
        II=K
70    CONTINUE
80  CONTINUE
    HM=ABS(HH)
    IF (HM.LE.HA) THEN
      A(M1)=-HM
      RETURN
    END IF
    A(M1)=HM
    HA=HM
```

```
      IM=IX(1)
      H1=HH
      J=1
      DO 100 I=1,N
        IF (I.EQ.IX(J)) THEN
          IF (J.LT.M1) J=J+1
        ELSE
          H2=A(M)
          DO 90 K=M-1,1,-1
90        H2=H2*X(I)+A(K)
          H2=H2-Y(I)
          IF (ABS(H2).GT.HM) THEN
            HM=ABS(H2)
            H1=H2
            IM=I
```

```
            END IF
         END IF
100      CONTINUE
      IF (IM.EQ.IX(1)) RETURN
      I=1
110      IF (IM.GE.IX(I)) THEN
         I=I+1
         IF (I.LE.M1) GOTO 110
      END IF

      IF (I.GT.M1) I=M1
      IF (I.EQ.(I/2)*2) THEN
         H2=HH
      ELSE
         H2=-HH
```

```
      END IF
      IF (H1*H2.GE.0.0) THEN
        IX(I)=IM
        GOTO 20
      END IF
      IF (IM.LT.IX(1)) THEN
        DO 120 J=M,1,-1
120        IX(J+1)=IX(J)
        IX(1)=IM
        GOTO 20
      END IF
      IF (IM.GT.IX(M1)) THEN
        DO 130 J=2,M1
130        IX(J-1)=IX(J)
        IX(M1)=IM
```

```
    GOTO 20
END IF
IX(I-1)=IM
GOTO 20
END
```